Math 2650

A. J. Meir

Laplace Transforms

In order to use Laplace transforms we must load the integral transforms package.

```
> restart:with(inttrans):
```

We compute some transforms of functions and some inverse transforms

```
> laplace(sinh(t),t,s);
```

$$\frac{1}{s^2-1}$$

```
> invlaplace(1/(s^2-1),s,t);
```

$$\sinh(t)$$

```
> laplace(t^2*sinh(t),t,s);
```

$$\frac{1}{(s-1)^3}-\frac{1}{(s+1)^3}$$

```
> invlaplace(1/((s-1)^3)-1/((s+1)^3),s,t);
```

$$t^2\sinh(t)$$

```
> laplace(Heaviside(t-3),t,s);
```

$$\frac{e^{-3s}}{s}$$

```
> invlaplace(exp(-3*s)/s,s,t);
```

$$\mathrm{Heaviside}(t-3)$$

Discontinuous Functions

Now lets look at some discontinuous functions

```
> f(t):=piecewise(0<=t and t<1, 1-t, 1<=t, 1);
```

$$f(t) := \begin{cases} 1-t & 0 \le t \textbf{ and } t < 1 \\ 1 & 1 \le t \end{cases}$$

```
> plot(f(t),t=0..5,discont=true);
```




```
> laplace(f(t),t,s);
```

$$\frac{1}{s}+\frac{-1+e^{-s}\,(s+1)}{s^2}$$

This is not too good! so lets try to help Maple along.

```
> g(t):=(1-t)*(Heaviside(t)-Heaviside(t-1))+Heaviside(t-1);
```

$$g(t) := (1-t)\,(\text{Heaviside}(t)-\text{Heaviside}(-1+t))+\text{Heaviside}(-1+t)$$

```
> plot(g(t),t=0..5,discont=true);
```




```
> laplace(g(t),t,s);
```

$$\frac{1}{s}+\frac{-1+e^{-s}\,(s+1)}{s^2}$$

```
> h(t):=Heaviside(sin(2*t));
```

$$h(t) := \text{Heaviside}(\sin(2\,t))$$

```
> plot(h(t),t=0..12,discont=true);
```




```
> laplace(h(t),t,s);
```

$$laplace(\text{Heaviside}(\sin(2\,t)), t, s)$$

```
> 1/(1-exp(-Pi*s))*int(exp(-s*t),t=0..Pi/2);
```

$$-\frac{-1+e^{-\frac{1}{2}\pi s}}{\left(1-e^{-\pi s}\right) s}$$

```
> invlaplace(-(exp(-1/2*Pi*s)-1)/((1-exp(-Pi*s))*s),s,t);
```

$$\frac{1}{2}+\frac{1}{2}\,(-1)^{\text{floor}\left(\frac{2\,t}{\pi}\right)}$$

Differential Equations

Consider the differential equation

$$\frac{d^2 x}{dt^2}+2\,\frac{dx}{dt}+x=f(t) \quad x(0)=0 \quad \frac{dx}{dt}(0)=0,$$

where $f(t)=\begin{cases} 1 & -t \le 0 \textbf{ and } t < 1 \\ 0 & 1 \le t \end{cases}$.

We define the right hand side, differential equation, and initial conditions.

```
> f(t):=Heaviside(t)-Heaviside(t-2);
```

$$f(t) := \text{Heaviside}(t) - \text{Heaviside}(t-2)$$

```
> de:=diff(x(t),t,t)+2*diff(x(t),t)+x(t)=f(t);
```

$$de := \frac{d^2}{dt^2}\,x(t) + 2\left(\frac{d}{dt}\,x(t)\right) + x(t) = \text{Heaviside}(t) - \text{Heaviside}(t-2)$$

```
> ic:={x(0)=0,D(x)(0)=0};
```

$$ic := \{D(x)(0)=0, x(0)=0\}$$

Laplace transform the differential equation.

```
> LEq:=laplace(de,t,s);
```

$$LEq := s^2\,laplace(x(t), t, s) - D(x)(0) - s\,x(0) + 2\,s\,laplace(x(t), t, s) - 2\,x(0)$$

$$+ \mathit{laplace}(x(t), t, s) = \frac{1 - e^{-2s}}{s}$$

Substitute the values for the initial conditions and solve the resulting algebraic equation for the transform of the solution.

```
> LEqIc:=subs(ic,LEq);
```

$$LEqIc := s^2\, \mathit{laplace}(x(t), t, s) + 2\, s\, \mathit{laplace}(x(t), t, s) + \mathit{laplace}(x(t), t, s) = \frac{1 - e^{-2s}}{s}$$

```
> LT:=solve(LEqIc,laplace(x(t),t,s));
```

$$LT := -\frac{-1 + e^{-2s}}{s\,(s^2 + 2\,s + 1)}$$

Invert to find the solution.

```
> invlaplace(LT,s,t);
Error, (in gcd/LinZip) input must be polynomials over the
integers
> invlaplace(1/(s*(s^2+2*s+1)),s,t)-invlaplace(exp(-2*s)/(s*
  (s^2+2*s+1)),s,t);
```

$$1 - e^{-t}\,(t + 1) - \mathrm{Heaviside}(t - 2)\,\left(1 - e^{-t+2}\,(-1 + t)\right) \qquad \textbf{(3.1)}$$

```
> sol:=%;
```

$$sol := 1 - e^{-t}\,(t + 1) - \mathrm{Heaviside}(t - 2)\,\left(1 - e^{-t+2}\,(-1 + t)\right)$$

And finally graph the solution.

```
> plot(sol,t=0..8);
```



Consider the differential equation

$$\frac{d^2 x}{dt^2} + 4\,x = f(t) \qquad x(0) = \frac{1}{2} \qquad \frac{dx}{dt}(0) = 0,$$

where $f(t) = \begin{cases} 4 - 2\,t & -t \le 0 \textbf{ and } t < 4 \\ 0 & 4 \le t \end{cases}$.

We define the right hand side, differential equation, and initial conditions.

```
> f(t) := piecewise(0 <= t and t < 4,4-2*t,4 <= t,0);
```

$$f(t) := \begin{cases} 4 - 2\,t & 0 \le t \textbf{ and } t < 4 \\ 0 & 4 \le t \end{cases}$$

Note: Maple can convert automatically to notation using the Heaviside function.

```
> f(t):=convert(%,Heaviside);
```

$$f(t) := 4\,\mathrm{Heaviside}(t) - 4\,\mathrm{Heaviside}(-4+t) - 2\,t\,\mathrm{Heaviside}(t) + 2\,t\,\mathrm{Heaviside}(-4+t)$$

```
> de:=diff(x(t),t,t)+4*x(t)=f(t);
```

$$de := \frac{d^2}{dt^2}\,x(t) + 4\,x(t) = 4\,\mathrm{Heaviside}(t) - 4\,\mathrm{Heaviside}(-4+t) - 2\,t\,\mathrm{Heaviside}(t) + 2\,t\,\mathrm{Heaviside}(-4+t)$$

```
> ic:={x(0)=1/2,D(x)(0)=0};
```

$$ic := \left\{\mathrm{D}(x)(0) = 0,\ x(0) = \frac{1}{2}\right\}$$

Laplace transform the differential equation.

```
> LEq:=laplace(de,t,s);
```

$$LEq := s^2\,laplace(x(t), t, s) - \mathrm{D}(x)(0) - s\,x(0) + 4\,laplace(x(t), t, s) = \frac{4}{s} + \frac{2\left(-1 + e^{-4s}\,(2\,s+1)\right)}{s^2}$$

Substitute the values for the initial conditions and solve the resulting algebraic equation for the transform of the solution.

```
> LEqIc:=subs(ic,LEq);
```

$$LEqIc := s^2\,laplace(x(t), t, s) - \frac{1}{2}\,s + 4\,laplace(x(t), t, s) = \frac{4}{s} + \frac{2\left(-1 + e^{-4s}\,(2\,s+1)\right)}{s^2}$$

```
> LT:=solve(LEqIc,laplace(x(t),t,s));
Warning, solutions may have been lost
```

$$LT :=$$

```
> LT:=(4/s+(2*(-1+exp(-4*s)*(2*s+1)))/s^2+s/2)/(s^2+4);
```

$$LT := \frac{\frac{4}{s} + \frac{2\left(-1 + e^{-4s}\,(2\,s+1)\right)}{s^2} + \frac{1}{2}\,s}{s^2+4} \qquad \textbf{(3.2)}$$

Invert to find the solution.

```
> invlaplace(LT,s,t);
Error, (in gcd/LinZip) input must be polynomials over the
```

integers

```
> invlaplace(4/(s*(s^2+4)),s,t)+invlaplace(2*(-1+exp(-4*s)*(2*s+1))/(s^2*(s^2+4)),s,t)+invlaplace(s/(2*(s^2+4)),s,t);
```

$$1 - \frac{1}{2}\cos(2\,t) - \frac{1}{2}\,t + \frac{1}{4}\sin(2\,t) + \frac{1}{4}\,\mathrm{Heaviside}(-4+t)\left(-8 + 8\sin(-4+t)^2 - \sin(-8+2\,t) + 2\,t\right) \tag{3.3}$$

```
> sol:=%;
```

$$sol := 1 - \frac{1}{2}\cos(2\,t) - \frac{1}{2}\,t + \frac{1}{4}\sin(2\,t) + \frac{1}{4}\,\mathrm{Heaviside}(-4+t)\left(-8 + 8\sin(-4+t)^2 - \sin(-8+2\,t) + 2\,t\right)$$

And finally graph the solution.

```
> plot(sol,t=0..9);
```